
Visaginas

Sprendimai (1 diena)

1. Mikroschema. 

(VIII–XII klasės). 

Sprendimas

Šis uždavinys nereikalauja ypatingo algoritmo, svarbu tinkamai parinkti duomenų struktūras.

Schema (ryšius) patogu saugoti įrašų masyve. Ryšį nusako įrašas:

Const C3 = 300;            { Maksimalus ryšių skaičius}
Type Ryšys = record 

               e1 : char;  n1 : integer;  { Ryšio pradžios elemento vardas ir numeris}
               e2 : char;  n2 : integer;  { Ryšio pabaigos elemento vardas ir numeris}
          end;

          Sar = array[1..C3] of Ryšys;  
Schema (ryšius) 

O kiekvieną elementą nusako toks įrašas: 

 type Elem = record        { Elemento aprašymas}
               d1, d2, r : integer; { Du įėjimai ir vienas išėjimas}

             end;

Analogiškai sukuriami kiekvieno tipo  elementų masyvai.

Pradžioje inicializuojami visi masyvai (elementų įėjimai ir išėjimai nežinomi) ir skaitant duomenis įsimenamos visos įėjimo signalų reikšmės. Taip pat visi ryšiai įsimenami ryšių masyve.

Pirmasis žingsnis – analizuojamas ryšių masyvas, randami visi įėjimo (P tipo) signalai ir jie siunčiami į atitinkamus elementus (t.y. užpildomos atitinkamų elementų įėjimo reikšmės).  Ryšiai su P tipo signalais pašalinami iš masyvo.

O toliau kol ryšių sąrašas netuščias kartojami tokie veiksmai:

· Peržiūrimi visi A tipo elementai: jeigu elemento išėjime signalas yra, tai ryšių sąraše ieškomi to elemento ryšiai su kitais elementais (t.y. užpildomos pastarųjų elementų įėjimų reikšmės), o šie ryšiai pašalinami iš masyvo;

· Analogiškai peržiūrimi I tipo elementai;

· Peržiūrimi visi elementai. Elementams, kurių abu įėjimai turi signalą, suskaičiuojamas išėjimo signalas;

Žemiau pateikta šį algoritmą realizuojanti programa:

program Mikroschema;

const

  Duom = 'schema.dat'; { Duomenų bylos vardas}
  Rez  = 'schema.rez'; { Rezultatų bylos vardas}
  C1 = 20;             { Maksimalus mikroschemos įėjimų ir išėjimų skaičius}
  C2 = 100;            { Maksimalus miokroschemos elementų skaičius}
  C3 = 300;            { Maksimalus ryšių skaičius}
type

  M1 = array[1..C1] of integer;  { Pradinio ir rezultatų signalo masyvo tipas}
  Elem = record        { Elemento aprašymas}
            d1, d2, r : integer; { Du įėjimai ir vienas išėjimas}
         end;

  M2 = array[1..C2] of Elem;    { Elementų sąrašo masyvų tipas}
  Rysys = record e1 : char;  n1 : integer;  { Ryšio pradžios elemento vardas ir numeris}
                 e2 : char;  n2 : integer;  { Ryšio pabaigos elemento vardas ir numeris}
          end;

  Sar = array[1..C3] of Rysys;  { Ryšių sąrašo masyvo tipas}
var  P,  G : M1;                { Mikroschemos įėjimo ir išėjimo signalai}
    nP, nG : integer;           { Įėjimo signalo ilgis nP ir išėjimo signalo ilgis nG}
     A,  I : M2;                { Elementų sąrašai pagal tipą. Kadangi elementai

               numeruojami eilės tvarka, tai jų vietą masyve nusako jų numeris schemoje.}
     R : Sar;  nR : integer;    { Ryšių sąrašas}
{---------------------------------------------}

{ Įėjimai ir išėjimai gali turėti tris reikšmes: yra signals 1, signalas 0 ir dar nėra signalo. Signalo nebuvimą koduojame –1. }

procedure Duomenys;

  var F : text;

      j, k : integer;

      sim  : char;   eil : string[2];

begin                                     { Pradinės masyvų reikšmės }
  for j := 1 to C1 do begin P[j] := -1;  G[j] := -1; end;

  for j := 1 to C2 do begin

    with A[j] do begin d1 := -1; d2 := -1; r := -1 end;

    with I[j] do begin d1 := -1; d2 := -1; r := -1 end;

  end;

  Assign(F, Duom); Reset(F);              { Paruošiama duomenų byla skaitymui}
  ReadLn(F, nP, nG);            { Skaitomos įėjimo signalo ir išėjimo signalo dydis}
  for j := 1 to nP do begin     { Skaitomas įėjimo signalas}
      Read(F, sim);

      if sim = '1' then P[j] := 1

                   else P[j] := 0;

  end;

  ReadLn(F);

  ReadLn(F, nR);                { Skaitomas ryšių skaičius}
  for j := 1 to nR do with R[j] do   { Skaitomi ryšiai}
                  { Elemento vardas, jo numeris, tarpas, elemento vardas ir jo numeris}

     ReadLn(F, e1, n1, sim, e2, n2 );     Close(F);

end;

{---------------------------------------------}

{ Įėjimo signalas siunčiamas schemos elementams ir, jeigu yra, į išėjimą}

procedure Pradzia;

  var j, k : integer;

begin

  j := 1;

  while j <= nR do begin    { Peržiūrimas ryšių sąrašas }
    if R[j].e1 = 'P' then   { Nagrinėjami tik mikroschemos įėjimo ryšiai}
    begin

      k := R[j].n1;

                             { Nagrinėjama,kur patenka signalas }

      if R[j].e2 = 'A' then with A[ R[j].n2 ] do { Kai patenka į A tipo elementą }
        if d1 = -1 then d1 := P[ k ]

                   else d2 := P[ k ];

      if R[j].e2 = 'I' then with I[ R[j].n2 ] do { Kai patenka į I tipo elementą }
        if d1 = -1 then d1 := P[ k ]

                   else d2 := P[ k ];

      if R[j].e2 = 'G' then G[ R[j].n2 ] := P[ k ];  { Kai patenka į išėjimą }
       R[j] := R[nR];  nR := nR - 1;  j := j - 1;{Išnagrinėtas ryšys šalimanas iš masyvo}
    end;

    j := j + 1;

  end;

end;

{---------------------------------------------}

{ Signalas siunčiamas iš elementų tipo A arba I:

  D – elenetų sąrašas, sim – elemento tipo vardas }
procedure Elementas(var D : M2;  sim : char);

  var j, k : integer;

begin

  for j := 1 to C2 do   begin { Elementų masyvo peržiūra }
    if D[j].r > -1 then begin { Jeigu elemento išėjime signalas jau yra, tai }
      k := 1;            { ryšių sąraše ieškomi to elemento ryšiai su kitais elementais}
      while k <= nR do  begin

        if (R[k].e1 = sim) and (R[k].n1 = j) then begin { Ar elementas tipo sim? }
          if R[k].e2 = 'A' then with A[ R[k].n2 ] do { Ryšys su A tipo elementu}
            if d1 = -1 then d1 := D[j].r

                       else d2 := D[j].r;

          if R[k].e2 = 'I' then with I[ R[k].n2 ] do  { Ryšys su I tipo elementu}
            if d1 = -1 then d1 := D[j].r

                       else d2 := D[j].r;

          if R[k].e2 = 'G' then G[ R[k].n2 ] := D[j].r; { Ryšys su mikroschemos išėjimu}
          R[k] := R[nR]; nR := nR - 1;  k := k - 1;   { Šalinamas ryšys iš masyvo }
        end;

        k := k + 1;

      end;

    end;

  end;

end;

{---------------------------------------------}

{ Elementams, kurių abu įėjimai turi signalą, skaičiuojamas išėjimo signalas }

  procedure Signalas;

    var j, k : integer;

  begin

    for j := 1 to C2 do begin

      with A[j] do                    { Elementas A tipo}
        if( d1 > -1) and (d2 > -1) then

          begin  k := d1 + d2;        { Arba operacijos rezultats }
                                      { Signalo neigimas išėjime  }

                 if k = 0 then r := 1

                          else r := 0;

          end;

      with I[j] do                    { Elementas I tipo}
        if( d1 > -1) and (d2 > -1) then

          begin  k := d1 * d2; ;      { Ir operacijos rezultats }
                                      { Signalo neigimas išėjime  }

                 if k = 0 then r := 1

                          else r := 0;

          end;

    end;

  end;

{---------------------------------------------}

procedure Darbas;

  var j : integer;

begin

  Signalas;             { Signalų elementų išėjime skaičiavimas po įėjimo signalo}
  while nR > 0 do begin { Kol ryšių sąrašas netuščias }
   Elementas( A, 'A');  { Elementų A tipo sąrašo peržiūra }
   Elementas( I, 'I');  { Elementų I tipo sąrašo peržiūra }
   Signalas;            { Signalų elementų išėjimams skaičiavimas }
  end;

end;

{---------------------------------------------}

{ Rezultato rašymas į bylą }

procedure Rezultatas;

   var j : integer; F : text;

 begin

   Assign(F, Rez); Rewrite(F);

   for j := 1 to nG do Write(F, G[j]:1);

   Close(F);

 end;

{---------------------------------------------}

begin

  Duomenys;     { Duomenų skaitymas iš bylos ir jų paruošimas darbui}
  Pradzia;      { Pradinio signalo siuntimas}
  Darbas;       { Mikroschemos nagrinėjimas }
  Rezultatas;   { Rezultato surašymas byloje}
end.

Galimas ir kitoks sprendimo būdas. Visi elementai saugomi viename masyve. Pirmosios 20 pozicijų rezervuotos įėjimo elementams, tolesnės 100 pozicijų rezervuotos A tipo elementams, dar tolesnės 100 pozicijų – I tipo elementams ir paskutiniosios 20 pozicijų – G tipo elementams.

Pavyzdžiui, elemento I2 numeris masyve būtų nr = 20+100+2=122.

Kiekvienas elementas aprašomas tokiu įrašu:

Type elementas = record

                     in1, in2,   { įėjimų elementų numeriai masyve }
                     r: integer; {  išėjimo reikšmė}
                     op: char; { operacija }
                   end;

Jei tai yra P tipo elementas, tuomet iš karto užpildomas jo reikšmės laukas (r), tačiau nebepildomi įėjimų numeriai.

Jei tai G tipo elementas, tuomet neužpildomas laukas in2. Nes išėjimo elementas turi tik vieną įėjimą.

Atkreipiame dėmesį, kad in1 ir in2 yra ne įėjimo elementų reikšmės, o įėjimo elementų numeriai masyve.

Suprantama, kad tokioje struktūroje saugant elementus ne visi masyvo elementai bus užpildyti. Tai priklausys nuo pradinių duomenų kiekio.

Tada išėjimų reikšmės apskaičiuojamos rekursiškai.  Suprantama, radus tarpinio elemento reikšmę, ji įsimenama, kad jos vėl prireikus netektų skaičiuoti iš naujo.

Žemiau pateikta programa, realizuojanti šį algoritmą.

program mikroschema;

    const PR = 'schema.dat';

        RZ = 'schema.rez';

        MP = 20;

        MG = 20;

        MA = 100;

        MI = 100;

        MAX_E = MP+MA+MI+MG; { maksimalus  elementų skaičius }
        MAX_R = 300;         { maksimalius ryšių skaičius }
  type elementas = record

                     in1, in2, r: integer; { įėjimų elementų numeriai masyve ir išėjimo reikšmė}
                     op: char; { operacija }
                   end;

       elem_sar = array [0..MAX_E] of elementas;

       rysys = record

                 op1, op2: char;

                 nr1, nr2: integer;

               end;

       rys_mas = array [1..MAX_R] of rysys;

  { žemiau aprašyti globalieji kintamieji }

  var el_sar: elem_sar;

  function skaiciuoti (ind: integer): integer;

    { rekursinė procedura, skaičiuojanti elemento išėjimo reikšmę }

     var r1, r2: integer;

  begin

    if el_sar[ind].r=-1 { jei reikšmė nežinoma }

       then begin

              r1 := skaiciuoti (el_sar[ind].in1);

              r2 := skaiciuoti (el_sar[ind].in2);

              case el_sar[ind].op of

              'A': if r1+r2=0

                      then el_sar[ind].r := 1

                      else el_sar[ind].r := 0;

              'I': if r1*r2 = 1

                      then el_sar[ind].r := 0

                      else el_sar[ind].r := 1;

              'G': el_sar[ind].r := r1;

              end;

            end;

      skaiciuoti := el_sar[ind].r;

   end; { skaičiuoti }
  procedure parengti (Rm: rys_mas; R: integer; { ryšiai, surašyti į simbolių masyvus }
                      Pdv: string; { įėjimų reikšmės }
                      var el_sar: elem_sar);

   { parengia duomenis, sukuria elementų masyvą }

    var tarpas, i, ind, nm1: integer;

  begin

    el_sar[0].r := 1; { fiktyvus elementas }
    for i := 1 to MAX_E do

     begin

       el_sar[i].r := -1; { elementu reikšmės nežinomos }

       el_sar[i].in1 := 0;

       el_sar[i].in2 := 0;

     end;

    for i := 1 to R do

      begin

        with Rm[i] do

          begin

            if (op1 = 'P') and (el_sar[nr1].r = -1)

               then if Pdv[nr1] = '0' { P elementų reikšmės žinomos }

                    then el_sar[nr1].r := 0

                    else el_sar[nr1].r := 1;

            case op2 of { apskaičiuojame elemento eilės numerį masyve }
              'A': ind := MP+nr2;

              'I': ind := MP+MA+nr2;

              'G': ind := MP+MA+MI+nr2;

            end;

            case op1 of { apskaičiuojame elemento eilės numerį masyve }
              'A': nm1 := MP+nr1;

              'I': nm1 := MP+MA+nr1;

              'G': nm1 := MP+MA+MI+nr1;

              'P': nm1 := nr1;

            end;

            el_sar[ind].op := op2;

            if el_sar[ind].in1 = 0

               then el_sar[ind].in1 := nm1

               else el_sar[ind].in2 := nm1

        end;

      end;

  end; { parengti }

  procedure skaityti (var P, G, R: integer; var Pdv: string;

                      var Rm: rys_mas);

…

  begin

…

  end;  { skaityti }

  procedure spausdinti (ind, G: integer);

…

  begin

…

  end; { spausdinti }

  var Rm: rys_mas;

      ind, i, R, P, G: integer;

      Pdv: string;

begin

  skaityti (P, G, R, Pdv, Rm);

  parengti (Rm, R, Pdv, el_sar);

  ind := MP+MA+MI;

  for i := ind+1 to ind + G do { skaičiuojamos visų išėjimo elementų reikšmės }

    skaiciuoti(i);

  spausdinti(ind, G);

end.

Testai

	Failas
	Duomenys
	Rezultatas

	Schema.dat
	11001
	010

	Schema1.dat
	1001
	100110

	Schema2.dat
	101010
	0

	Schema3.dat
	10
	10000011

	Schema4.dat
	11
	0011100

	Schema5.dat
	110011110001101001110101010
	01100101011111


	Schema.dat
	Užduoties pavyzdys

	Schema1.dat
	Nėra nei vieno elemento A, I. Įėjimo signalai patenka tiesiogiai į išėjimus

	Schema2.dat
	Nedidelė paprasta schemutė; tik vienas išėjimas

	Schema3.dat
	Nuosekliai jungiami 8 vidiniai elementai. Išėjimai iš skirtingų schemos vietų.

	Schema4.dat
	5 nuoseklūs ir 6 lygiagretūs jungimai. Išėjimai tik iš paskutinės pakopos elementų

	Schema5.dat
	“Sudėtinga” schema iš 27 iėjimų, 14 išėjimų, 19 A tipo ir 17 I tipo elementų. Jungimai sudėtingi.







2. Lietsargiai. 

(VIII–IX klasės). 

Sprendimas

Tai pats lengviausias olimpiados uždavinys. Jo sprendimas paprastas:

Pradžioje laikome, kad visi sausi, t.y. sušlapusių žmonių skaičius lygus 0. Imame kiekvieną žmogų be lietsargio ir peržiūrime visus žmones su lietsargiais. Jei nė vienas lietsargis jo neuždengia, padidiname sušlapusių žmonių skaičių vienetu.

Primename atstumo tarp dviejų taškų, kurių koordinatės (x1, y1) ir (x2, y2) formulę: 

  const  MAX = 100; { maksimalus žmonių skaičius }

  type zmogus = record
                  x, y, d: real; { koord. ir lietsargio diametras }
                end;

       lentele = array [1..MAX] of zmogus;

  function uzstos (zl, zb: zmogus): boolean;

   { ar žmogus su lietsargiu užstos nuo lietaus žmogų be lietsargio }

  begin

     uzstos := (sqrt(sqr(zl.x - zb.x) + sqr(zl.y - zb.y)) <= zl.d;

  end; { užstos }
  function suslaps (N, M: integer; su, be: lentele): integer;

    { kiek žmonių sušlaps }

    var j, i, slapi: integer;

        sausas: boolean;

  begin

    slapi := 0;

    for i := 1 to M do { be lietsargių }
      begin

        j := 1; sausas := false;

        while (j <= N) and not sausas do { ieško žmonių su lietsargiais }
          begin

            sausas := uzstos (su[j], be[i]);

            j := j + 1;

          end;

        if not sausas

           then slapi := slapi + 1;

      end;

    suslaps := slapi;

  end; { sušlaps }
Testai

0 testas – pavyzdys, Ats. 3;

1 testas – atsitiktinis, Ats. 5;

2 testas – visi be skėčių sušlaps, Ats. 10;

3 testas – visi be skėčių liks sausi, Ats. 0;

4 testas – 2 žmonės be lietsargių pateks ant apskritimo krašto, Ats. 4;

5 testas – maksimalus skaičius duomenų, Ats. 57;

2. Susisiekimas traukiniais.

X-XII klasės

Sprendimas

Traukinių skaičius bei traukinių maršrutai (keliose stotelėse stoja traukinys, reikia persėsti ar ne) šio uždavinio sprendimui neturi jokios reikšmės. Reisu laikysime įrašą, kuriame saugoma tokia informacija:

· traukinio išvykimo stotis A;

· traukinio atvykimo stotis B;

· išvykimo iš stoties A laikas;

· atvykimo į stotį B laikas;

Tad skaitant pradinius duomenis visiškai nereikia saugoti traukinių skaičiaus ir kiekvieno traukinio maršruto. Svarbu tik įsiminti visus reisus. 

Sąlygoje prašoma surasti važiavimo iki paskutiniosios stotelės laiką minutėmis, tad galima jau skaitant duomenis išvykimo/atvykimo laikus paversti minutėmis ir programoje naudoti tik minutes.

Taip pat prašoma surasti tik tokius maršrutus (t.y. jų laiką), kai atvykstama iki vidurdienio, tai skaitant duomenis galima neįsiminti (arba vėliau pašalinti) visus reisus, kur išvykstama arba atvykstama po vidurdienio. 

Tolesniam darbui bus patogu, jei visus reisus surikiuosime išvykimo laiko didėjimo tvarka. T.y. kuo ankstesnis reiso išvykimo laikas, tuo arčiau eilės pradžios įrašytas tas reisas.

Tegul t – laikas, praėjęs nuo 0 val. 0 min. išreikštas minutėmis. Kiekvienai stočiai k priskirkime svorį W(t, k) – mažiausią skaičių stočių, kurias reikia pervažiuoti, norint pasiekti stotelę k per laiko momentą t, kai važiuojama nuo pirmos stoties. 

Pirmiausia užpildome pirmąją lentelės eilutę, kai k (1: Wt(t, 1) = 1, nes pirmąją stotelę keleivis visada „pervažiuos“ nuliniu laiko momentu. Po to inicializuojame 
W(t, k) = ∞, kai S≥k>1 (realizacijoje vietoje ∞ naudojame, pavyzdžiui, maxlongint ).

Atliekame iteracijas pagal laiką t (t=0,1, 2,3,...). 

Atliekant vieną iteraciją peržiūrime visus reisus, išvykstančius laiko momentu t. Pirmiausia patikrinama, ar W(t, k) < ∞, (t.y. ar į tą stotį buvo galima atvykti iki laiko momento t). Jei išvykti galima, tuomet atnaujinamas atvykimo stoties svoris. Sakykime, kad reiso atvykimo laikas – mn, atvykimo stotis – st (išvykimo laikas ir stotis atitinkamai t ir k). 

Tuomet: 
   W(mn,st) := min {W( mn, st ), W(t,k) + 1}

Iteracijas atliekame tol, kol  t ≤ 720 ir kol W(t, S)= ∞.

Testai

	Testo nr.
	Prad. duom

T ir S
	Rezultatas
	Taškai
	Komentaras

	0a
	1 3
	40 3
	1
	Sąlygos testai

	0b
	1 3
	NE
	1
	

	0c
	3 5
	392 3
	2
	

	1
	10 10
	11 2
	3
	Nesudėtingi testai, visi jie sugeneruoti  automatiškai ( panaudotas atsitiktinių  skaičių generatorius)

	2
	21 100
	486 4
	3
	

	3
	311 1000
	637 7
	3
	

	4
	351 2500
	403 14
	3
	

	5
	956 5000
	635 15
	4
	Vidutinio sunkumo testai. Irgi sugeneruoti automatiškai.

	6
	554 7500
	501 24
	4
	

	7
	181 1000
	318 10
	5
	

	8
	771 10000
	503 22
	8
	Sudėtingi testai, daugumos kurių paskirtis – patikrinti parašytos programos efektyvumą.

	9
	1000 10000
	513 12
	8
	

	10
	1000 10000
	496 33
	8
	

	11
	909 10000
	NE
	7
	Vienintelis testas, kuriame atsakymas NE. Pakankamai sudėtingas.

	12
	1000 8002
	380 7
	8
	Įvairaus sunkumo testai, kurie sukurti „rankiniu“ būdu, daugumos jų  pagrindinė paskirtis – patikrinti, ar gerai skaičiuojamas aplankytų stotelių skaičius.

	13
	10 982
	380 100
	5
	

	14
	4 7
	378 2
	4
	

	15
	1 6
	720 5
	3
	


3. Vagonai.

X-XII klasės.

Sprendimas

Uždavinys sprendžiamas dinaminio programavimo metodu. Sakykime, turime traukinį iš dvylikos vagonų:

V1  V2  V3  V4  V5  V6  V7  V8  V9  V10  V11  V12

Taip pat yra keturi užsakovai. Kiekvienam jų reikia palikti ne mažiau kaip 2 vagonus. 

Žemiau pateikta keletas galimų paskirstymų. 

V1 V2 ( V3 V4 ( V5 V6 ( V7 V8 V9 V10 V11 V12

V1 V2 V3 ( V4 V5 V6 ( V7 V8 V9 ( V10 V11 V12

V1 V2 V3 V4 V5 ( V6 V7 V8 ( V9 V10 ( V11 V12 

Galbūt vaizdžiau skambės pasakymas, kad juostelę su nupieštais vagonais reikia perkirpti U-1 kartą arba sukarpyti į U dalių taip, kad kiekvienoje atkirptoje dalyje būtų nupiešti bent du vagonai.

Aukščiau pateiktame pavyzdyje turime 4 užsakovus ir 12 vagonų. Panagrinėkime, kiek vagonų galime palikti paskutiniam (ketvirtajam) klientui. Galimi tik tokie kirpimai, nes mažiausiai šešis vagonus teks palikti pirmiesiems klientams, o mažiausiai du vagonai turi tekti paskutiniam klientui.

V1 V2 V3 V4 V5 V6 ( V7 V8 V9 V10 V11 V12

V1 V2 V3 V4 V5 V6 V7 ( V8 V9 V10 V11 V12

V1 V2 V3 V4 V5 V6 V7 V8 ( V9 V10 V11 V12

V1 V2 V3 V4 V5 V6 V7 V8 V9 ( V10 V11 V12

V1 V2 V3 V4 V5 V6 V7 V8 V9 V10 ( V11 V12

Sakykime, kad kerpame tarp 10 ir 11 vagono.

Tokiu atveju paskutinieji du vagonai atitenka ketvirtam užsakovui. Nesunkiai galima patikrinti, ar įvykdytas ketvirtojo kliento užsakymas ir, jei ne, – kiek užsakytų vagonų atiteko paskutiniajam užsakovui. 

Toliau mums lieka 10 pirmųjų vagonų ir 3 užsakovai. Reikia spręsti tiksliai tą patį uždavinį tik su mažesniais parametrais. Jei žinotume šio mažesnio uždavinio sprendinį, pridėję paskutinio užsakovo duomenis (arba patenkintą užsakymą arba vagonų skaičių) sužinotume, kiek daugiausia užsakymų įvykdyta bei atiduota tinkamų vagonų kerpant tarp 10 ir 11 vagonų. 

Kaip matėme, kirpti galima net penkiose vietose, tad reiktų išnagrinėti visus galimus kirpimo atvejus ir išrinkti geriausią variantą.

Visi aukščiau aprašyti samprotavimai veda dinaminio programavimo algoritmo link.

	
	Vagonai (V)

	Užsakovai (U)
	
	
	
	
	…
	

	
	
	
	
	
	
	

	
	
	
	
	
	
	

	
	
	
	
	
	
	

	
	…
	
	
	
	
	


Sprendimo metu užpildoma lentelė (U(V) iš tiek stulpelių, kokio ilgio yra traukinys (V), ir iš tiek eilučių, kiek yra užsakovų (U). 

Šios lentelės eilutės I ir stulpelio J sankirtoje Lent[I, J] bus įrašytas skaičius reiškiantis ieškomą sprendinį, jeigu turėtume I užsakovų, o traukinys turėtų J vagonų. Kitaip sakant, jei J-asis vagonas būtų paskutinysis I-ajam klientui atitekęs vagonas (kerpama tarp J-ojo ir (J+1)-ojo vagono). 

Lentelė pildoma tokiu būdu.

Galima pirmiausia užpildyti pirmąją lentelės eilutę Lent[1, J]. T.y. išnagrinėti atvejį, kai pirmajam užsakovui paliekama pirmieji J sąstato vagonų. Jei tarp pirmųjų J vagonų yra visi pirmojo kliento užsakyti vagonai ir J(2, tuomet Lent[1, J] ( (1, 0); Priešingu atveju (kai J(1 arba klientas gaus ne visus užsakytus vagonus), 
Lent[1, J] ( (0, z), kur z – pirmojo kliento užsakytų vagonų skaičius tarp pirmųjų J vagonų. 

Atkreipiame dėmesį, kad kiekvienas lentelės elementas yra įrašas iš dviejų laukų: įvykdytų užsakymų skaičiaus ir ieškomo vagonų skaičiaus. 

Reikia nepamiršti, kad pirmajam klientui galima palikti ne daugiau kaip 2((U-1) vagonų. Nes likusiems U-1 klientui reikia palikti bent po du vagonus. 

Sakykime, kad yra užpildyta (i-1) lentelės eilučių. Reikia pildyti i–ąją lentelės eilutę. Pavyzdžiui pildome laukelį Lent[i, j]. T.y. paskutiniojo i-ajam klientui paliekamo vagono numeris yra j. 

Pirmojo šiam klientui paliekamo vagono numeris (kita kirpimo riba) gali kisti nuo nr1(2*(i-1)+1 iki nr2(j-1. Išnagrinėjame kiekvieną atvejį atskirai ir surandame, ar įvykdytas užsakymas ir jei ne – kiek vagonų atiteks šiam užsakovui. 

Taigi, Lent[i, j]( max {Lent[I-1, nr-1] + (užs, vag), nr ([nr1, nr2]}

Lentelės pildymo pavyzdys.

Sakykime, turime tokius pradinius duomenis:

	Pradiniai duomenys
	Paaiškinimai

	4 7 
	Yra 4 skirtingų rūšių prekių, 7 vagonų sąstatas.

	1 2 4 3 4 1 3 
	Sąstato sudėtis.

	3 
	Yra 3 klientai.

	2 1 1 
	Pirmasis klientas užsisakė du 1 rūšies prekių vagonus.

	2 3 4 
	Antrasis, kaip ir trečiasis, klientas tikisi gauti po vagoną 3 ir 4 rūšies prekių.

	2 3 4
	


Pirmajam klientui galima palikti tokias sąstato dalis (prisimenant, kad jam reikia pristatyti ne mažiau 2 vagonų ir kitiems reikia palikti po ne mažiau 2 vagonus):

	
	Vagonai (V)
	Rezultatai

	Variantai
	1
	2
	4
	3
	4
	1
	3
	

	
	xx
	xx
	
	
	
	
	
	Abiem atvejais nepristatomas pilnas užsakymas, o tik 1 užsakytų prekių vagonas.

	
	xx
	xx
	xx
	
	
	
	
	


Pildymas antram klientui:

	
	Vagonai (V)
	Rezultatai

	
	1
	2
	4
	3
	4
	1
	3
	

	Variantai
	
	0 1
	0 1
	
	
	
	
	Ankstesnio veiksmo rezultatas

	
	
	
	xx
	xx
	
	
	
	(1, 0) + (0, 1)

	
	
	
	xx
	xx
	xx
	
	
	(1, 0) + (0, 1)

	
	
	
	
	xx
	xx
	
	
	(1, 0) + (0, 1)


Pildymas trečiam klientui:

	
	Vagonai (V)
	Rezultatai

	
	1
	2
	4
	3
	4
	1
	3
	

	Variantai
	
	
	
	1 1
	1 1
	
	
	Ankstesnio veiksmo rezultatas

	
	
	
	
	
	xx
	xx
	xx
	(1, 0) + (1, 1)

	
	
	
	
	
	
	xx
	xx
	(0, 1) + (1, 1)


	
	
	Vagonai (V)

	
	
	1
	2
	4
	3
	4
	1
	3

	Klientai (3)
	1
	
	0 1
	0 1
	
	
	
	

	
	2
	
	
	
	1 1
	1 1
	
	

	
	3
	
	
	
	
	
	
	2 1


Taip atrodytų visa užpildyta lentelė, o jos apatiniame dešiniajame kampe yra įrašytas ieškomas sprendinys.

Ieškomas rezultatas: 2 1

Testai

	Nr.
	Pradiniai duomenys
	Rezultatas
	Paaiškinimai

	1
	3 4 

1 2 3 2 

1 

2 3 3
	0 1
	Labai paprastas testas: vienas klientas, todėl jam viską ir nuvežam, tereikia paskaičiuoti, ar visų pageidaujamų prekių turima.

	2
	4 9 

1 4 3 1 2 3 2 4 4 

3 

2 2 1 

2 1 1 

2 2 2
	1 2
	Testo esmė, kad nereikia stengtis pilnai išpildyti pirmųjų dviejų klientų užsakymų.

	3
	10 100 

... 

50 

...
	3 20
	Vagonų kiekis = Užsakovų skaičius ( 2. Tereikia paskaičiuoti, kaip prekės pasiskirsto užsakovams.

	4
	4 100 

1 1 ... 1 1 2 2 1 ... 1

1 4 

8 

4 3 3 3 3 

4 3 3 3 3 

4 3 3 3 3 

2 2 2 

4 3 3 3 3 

4 3 3 3 3 

4 3 3 3 3 

2 2 2
	1 0
	Atvežama tik vienam užsakovui reikalingų prekių.

	5
	8 100 

1 ... 1 2 2 3 3 4 4 5 5

6 6 7 7 8 8 

8 

2 1 1 

2 2 2 

2 3 3 

2 4 4 

2 5 5 

2 6 6 

2 7 7 

2 8 8
	8 0
	Pristatomi visi užsakymai. Reikiamos prekės yra sąstato pabaigoje.

	6
	9 100 

9 9 ... 9 9 

50
	0 0
	Nepristatomas nė vienas užsakymas. 

Užsakymų aprašuose prekės iš rėžio 1 - 8

	7
	12 100 

...
	6 17
	Atsitiktinai sugeneruotas testas

	8
	9 100 

...
	5 37
	Atsitiktinai sugeneruotas testas

